
NAG Library Routine Document

F08FAF (DSYEV)

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

F08FAF (DSYEV) computes all the eigenvalues and, optionally, all the eigenvectors of a real n by n
symmetric matrix A.

2 Specification

SUBROUTINE F08FAF(JOBZ, UPLO, N, A, LDA, W, WORK, LWORK, INFO)

INTEGER N, LDA, LWORK, INFO

double precision A(LDA,*), W(*), WORK(max(1,LWORK))
CHARACTER*1 JOBZ, UPLO

The routine may be called by its LAPACK name dsyev.

3 Description

The symmetric matrix A is first reduced to tridiagonal form, using orthogonal similarity transformations,
and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally)
the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users’ Guide (3rd Edition) SIAM,
Philadelphiahttp://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University Press,
Baltimore

5 Parameters

1: JOBZ – CHARACTER*1 Input

On entry: if JOBZ ¼ N , compute eigenvalues only.

If JOBZ ¼ V , compute eigenvalues and eigenvectors.

Constraint: JOBZ ¼ N or V .

2: UPLO – CHARACTER*1 Input

On entry: if UPLO ¼ U , the upper triangular part of A is stored.

If UPLO ¼ L , the lower triangular part of A is stored.

Constraint: UPLO ¼ U or L .

3: N – INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N � 0.

4: AðLDA,�Þ – double precision array Input/Output

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the n by n symmetric matrix A.

If UPLO ¼ U , the upper triangular part of A must be stored and the elements of the array
below the diagonal are not referenced.

If UPLO ¼ L , the lower triangular part of A must be stored and the elements of the array
above the diagonal are not referenced.

On exit: if JOBZ ¼ V , then if INFO ¼ 0, A contains the orthonormal eigenvectors of the matrix A.

If JOBZ ¼ N , then on exit the lower triangle (if UPLO ¼ L ) or the upper triangle (if UPLO ¼ U )
of A, including the diagonal, is destroyed.

5: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08FAF
(DSYEV) is called.

Constraint: LDA � max 1;Nð Þ.

6: Wð�Þ – double precision array Output

Note: the dimension of the array W must be at least max 1;Nð Þ.
On exit: if INFO ¼ 0, the eigenvalues in ascending order.

7: WORKðmax 1;LWORKð ÞÞ – double precision array Workspace

On exit: if INFO ¼ 0, WORKð1Þ contains the minimum value of LWORK required for optimal
performance.

8: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FAF
(DSYEV) is called.

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the optimal size of the
WORK array, returns this value as the first entry of the WORK array, and no error message related
to LWORK is issued.

Suggested value: for optimal performance, LWORK � nbþ 2ð Þ � N, where nb is the optimal block
size for F08FEF (DSYTRD).

Constraint: LWORK � max 1; 3� N� 1ð Þ.

9: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution of
the program is terminated.

INFO > 0

If INFO ¼ i, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal
form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix Aþ Eð Þ, where
Ek k2 ¼ O �ð Þ Ak k2,

and � is the machine precision. See Section 4.7 of Anderson et al. (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n3.

The complex analogue of this routine is F08FNF (ZHEEV).

9 Example

This example finds all the eigenvalues and eigenvectors of the symmetric matrix

A ¼

1 2 3 4
2 2 3 4
3 3 3 4
4 4 4 4

0
BB@

1
CCA,

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

* F08FAF Example Program Text
* Mark 21 Release. NAG Copyright 2004.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NB, NMAX
PARAMETER (NB=64,NMAX=10)
INTEGER LDA, LWORK
PARAMETER (LDA=NMAX,LWORK=(NB+2)*NMAX)

* .. Local Scalars ..
DOUBLE PRECISION EERRBD, EPS
INTEGER I, IFAIL, INFO, J, LWKOPT, N

* .. Local Arrays ..
DOUBLE PRECISION A(LDA,NMAX), RCONDZ(NMAX), W(NMAX), WORK(LWORK),

+ ZERRBD(NMAX)
* .. External Functions ..

DOUBLE PRECISION X02AJF
EXTERNAL X02AJF

* .. External Subroutines ..
EXTERNAL DDISNA, DSYEV, X04CAF

* .. Intrinsic Functions ..
INTRINSIC ABS, INT, MAX

* .. Executable Statements ..
WRITE (NOUT,*) ’F08FAF Example Program Results’
WRITE (NOUT,*)

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read the upper triangular part of the matrix A from data file
*

READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
*
* Solve the symmetric eigenvalue problem
*

CALL DSYEV(’Vectors’,’Upper’,N,A,LDA,W,WORK,LWORK,INFO)
LWKOPT = INT(WORK(1))

*
IF (INFO.EQ.0) THEN

*
* Print solution
*

WRITE (NOUT,*) ’Eigenvalues’
WRITE (NOUT,99999) (W(J),J=1,N)

*
IFAIL = 0
CALL X04CAF(’General’,’ ’,N,N,A,LDA,’Eigenvectors’,IFAIL)

*
* Get the machine precision, EPS and compute the approximate
* error bound for the computed eigenvalues. Note that for
* the 2-norm, max( abs(W(i)) ) = norm(A), and since the
* eigenvalues are returned in ascending order
* max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)))
*

EPS = X02AJF()
EERRBD = EPS*MAX(ABS(W(1)),ABS(W(N)))

*
* Call DDISNA (F08FLF) to estimate reciprocal condition
* numbers for the eigenvectors
*

CALL DDISNA(’Eigenvectors’,N,N,W,RCONDZ,INFO)
*
* Compute the error estimates for the eigenvectors
*

DO 20 I = 1, N
ZERRBD(I) = EERRBD/RCONDZ(I)

20 CONTINUE
*
* Print the approximate error bounds for the eigenvalues
* and vectors
*

WRITE (NOUT,*)
WRITE (NOUT,*) ’Error estimate for the eigenvalues’
WRITE (NOUT,99998) EERRBD
WRITE (NOUT,*)
WRITE (NOUT,*) ’Error estimates for the eigenvectors’
WRITE (NOUT,99998) (ZERRBD(I),I=1,N)

ELSE
WRITE (NOUT,99997) ’Failure in DSYEV. INFO =’, INFO

END IF
*
* Print workspace information
*

IF (LWORK.LT.LWKOPT) THEN
WRITE (NOUT,*)
WRITE (NOUT,99996) ’Optimum workspace required = ’, LWKOPT,

+ ’Workspace provided = ’, LWORK
END IF

ELSE
WRITE (NOUT,*) ’NMAX too small’

END IF
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
99996 FORMAT (1X,A,I5,/1X,A,I5)

END

9.2 Program Data

F08FAF Example Program Data

4 :Value of N

1.0 2.0 3.0 4.0
2.0 3.0 4.0

3.0 4.0
4.0 :End of matrix A

9.3 Program Results

F08FAF Example Program Results

Eigenvalues
-2.0531 -0.5146 -0.2943 12.8621

Eigenvectors
1 2 3 4

1 0.7003 -0.5144 -0.2767 0.4103
2 0.3592 0.4851 0.6634 0.4422
3 -0.1569 0.5420 -0.6504 0.5085
4 -0.5965 -0.4543 0.2457 0.6144

Error estimate for the eigenvalues
1.4E-15

Error estimates for the eigenvectors
9.3E-16 6.5E-15 6.5E-15 1.1E-16

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

